Chapter 6
Multi-Period Optimal Power Flow

This chapter provides a solution for optimal power flow OPF problem in GAMS. Different OPF models are investigated, such as single and multi-period DC-AC optimal power flow.

6.1 Single Period Optimal DC Power Flow

There are some necessary conditions that make the DC power flow acceptable as an approximate solution for AC power flow such as:

- The ratio of $\frac{x_{ij}}{r_{ij}}$ should be large enough that r_{ij} can be neglected.
- The voltage magnitudes are approximately 1 pu.

The DC power flow concept for a two-bus network is shown in Fig. 6.1. The basic variables in DC power flow are voltage angles δ_i. The angle of the slack bus is assumed to be zero as the reference for the rest of network.

The technical and economic characteristics of generating units shown in Table 6.1 are given as follows:

The demand at bus 2 is $L_2 = 400$ MW, line reactance is $X_{12} = 0.2$ pu, and line flow limit is $P_{12}^{\max} = 1.50$ pu (per unit on 100 MVA base). The optimization problem which should be solved is formulated in (6.1).

$$\min_{P_g,\delta_i} \text{OF} = \sum_{g_1,g_2} a_g (P_g)^2 + b_g P_g + c_g \tag{6.1a}$$

$$P_{ij} = \frac{\delta_1 - \delta_2}{X_{12}} \tag{6.1b}$$

$$P_{g_1} = P_{12} \tag{6.1c}$$

$$P_{g_2} + P_{12} = L_2 \tag{6.1d}$$

A. Soroudi, *Power System Optimization Modeling in GAMS*,
DOI 10.1007/978-3-319-62350-4_6



Fig. 6.1 DC power flow for a two-bus network

GCode 6.1 The OPF GAMS code for two-bus network, Example (6.1)

```
Sets
Gen /g1*g2/
bus /1*2/;
Scalars
L2     /400/
X12  /0.2/
Sbase  /100/
P12_max  /1.5/  ;
Table  data(Gen,*)
      a      b       c        Pmin  Pmax
G1    3      20      100      28    206
G2    4.05   18.07   98.87    90    284;
Variables  P(gen),OF, delta(bus),P12;
Equations
eq1,eq2,eq3,eq4;

eq1  ..  OF=e=sum(gen,data(gen,'a')*P(gen)*P(gen)+data(gen,'b')
*P(gen)+data(gen,'c'));
eq2  ..  P('G1')=e=P12;
eq3  ..  P('G2')+P12=e=L2/Sbase;
eq4  ..  P12=e=(delta('1')-delta('2'))/X12;
P.lo(gen)=data(gen,'Pmin')/Sbase;
P.up(gen)=data(gen,'Pmax')/Sbase;
P12.lo=-P12_max;
P12.up=+P12_max;
delta.fx('1')=0;
Model  OPF  /all/;
Solve  OPF  us  qcp  min  of;
```

$$-P_{12}^{\max} \leq P_{12} \leq P_{12}^{\max} \tag{6.1e}$$

$$\delta_1 = 0 \quad \text{Slack} \tag{6.1f}$$

The GAMS code for solving the (6.1) is provided in GCode 6.1.

The operating costs would be \$306.108, $P_{12} = 150$ MW and $\delta_2 = -0.3$ (rad).

The general power flow concept is shown in Fig. 6.2. As it can be seen in Fig. 6.2, every bus might host some generation and demand. Each bus might be connected

Table 6.1 The techno-economic data of thermal units in two-bus OPF example

g	a_g (\$/MW^2)	b_g (\$/MW)	c_g (\$)	P_g^{min} (MW)	P_g^{max} (MW)
g_1	0.12	14.8	89	28	200
g_2	0.17	16.57	83	20	290



Fig. 6.2 General power flow concept

to other network buses by multiple branches (with different characteristics). The power balance between generation, demand and power transfers should be satisfied at every bus of the network under study.

The general mathematical formulation of OPF is described in (6.2). It should be noted that the technical terms of (6.2) are all linear. The only nonlinear part is the cost function. This makes the DC-OPF a strong tool for power system studies at the transmission level. This means that if the cost term can be expressed in linear form then the DC-OPF would become a linear programming problem and can be solved using linear solvers in GAMS such as CPLEX [1].

$$\mathrm{OF} = \sum_{g \in \Omega_G} a_g (P_g)^2 + b_g P_g + c_g \tag{6.2a}$$

$$P_{ij} = \frac{\delta_i - \delta_j}{x_{ij}} \quad ij \in \Omega_\ell \tag{6.2b}$$

$$\sum_{g \in \Omega_G^i} P_g - L_i = \sum_{j \in \Omega_\ell^i} P_{ij} : \lambda_i \quad i \in \Omega_B \tag{6.2c}$$

$$-P_{ij}^{max} \le P_{ij} \le P_{ij}^{max} \quad ij \in \Omega_\ell \tag{6.2d}$$

$$P_g^{min} \le P_g \le P_g^{max} \tag{6.2e}$$



Fig. 6.3 Three bus network example (**a**) data and (**b**) power flow solution

6.1.1 Three-Bus Network DC-OPF

The three-bus network data is shown in Fig. 6.3. This example is taken from [2] (Example 4B, p. 110).

The GAMS code for solving this problem is given as GCode 6.2: The developed code for solving the OPF problem in three-bus network (Fig. 6.3) is explained here. It is general and can be used for any network with any size.

- Three sets are defined: bus (all network buses), slack(bus) which shows slack buses with reference angle values, Gen (set of generating units)
- A scalar value named *Sbase* is defined for per unit calculations
- The set node is defined as the similar set to set *bus*
- The table *GenData* defines the technical and economic characteristics of generating units
- The set *GBconect* defines the connection point of each generating unit
- The table *BusData* specifies the demand values in each bus
- The set *conex* specifies how each bus is connected to the other network buses
- The table *branch* defines the branch characteristics
- Four variables are used for this formulation namely: *OF* (objective function), *Pij* (active power flow between bus and node), *Pg*(generating schedule of each generating unit), and *delta* (voltage angle at each bus)
- Three equations are defined: *const1* (active flow calculation between each pair of connected buses), *const2* (nodal active power balance in each bus), and *const3* (objective function calculation)
- The definition of the model *loadflow* (specifies for GAMS that which constraints should be taken into account)
- Variables’ limits specification

- Solve statement
- Generating report from the solved model
- The marginal value of *const2.m* provides the LMP at each bus. This is used for congestion cost calculation.

GCode 6.2 The DC-OPF GAMS code for three-bus network, Example (Sect. 6.1.1)

```
Sets
bus   /1*3/
slack(bus) /3/
Gen  /g1*g3/;
scalars
Sbase /100/;
alias(bus,node);
Table GenData(Gen,*)  Generating units characteristics
   b     pmin pmax
g1 10    0    65
g2 11    0    100;
* ----------------------------------------------------------------
set GBconect(bus,Gen) connectivity index of each generating unit
    to each bus
/1     .     g1
 3     .     g2   /  ;
Table BusData(bus,*) Demands of each bus in MW
          Pd
2         100;
set conex           Bus connectivity matrix
/
1     .     2
2     .     3
1     .     3/;
conex(bus,node)$(conex(node,bus))=1;
Table branch(bus,node,*)     Network technical characteristics
                  x          Limit
1     .     2     0.2        100
2     .     3     0.25       100
1     .     3     0.4        100  ;
branch(bus,node,'x')$(branch(bus,node,'x')=0)=branch(node,bus,'x'
    );
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node
    ,bus,'Limit');
branch(bus,node,'bij')$conex(bus,node) =1/branch(bus,node,'x');
Variables
OF
Pij(bus,node)
Pg(Gen)
delta(bus);
Equations const1,const2,const3;
const1(bus,node)$conex(bus,node).. Pij(bus,node)=e=
                                  branch(bus,node,'bij')*(delta(bus)-
                                      delta(node));
```

```
const2(bus) .. +sum(Gen$GBconect(bus,Gen),Pg(Gen))-BusData(bus,'
    pd')/Sbase=e=
                        +sum(node$conex(node,bus),Pij(bus,node)
                        );
const3 .. OF=g=sum(Gen,Pg(Gen)*GenData(Gen,'b')*Sbase);

Model loadflow /const1,const2,const3/;
Pg.lo(Gen)=GenData(Gen,'Pmin')/Sbase;
Pg.up(Gen)=GenData(Gen,'Pmax')/Sbase;
delta.up(bus)=pi; delta.lo(bus)=-pi; delta.fx(slack)=0;
Pij.up(bus,node)$((conex(bus,node)))=1* branch(bus,node,'Limit')/
    Sbase;
Pij.lo(bus,node)$((conex(bus,node)))=-1*branch(bus,node,'Limit')/
    Sbase;
Solve loadflow minimizing OF using lp;
parameter report(bus,*),Congestioncost;
report(bus,'Gen(MW)')= sum(Gen$GBconect(bus,Gen),Pg.l(Gen))*sbase
    ;
report(bus,'Angle')=delta.l(bus);
report(bus,'load(MW)')= BusData(bus,'pd');
report(bus,'LMP($/MWh)')=const2.m(bus)/sbase ;
Congestioncost= sum((bus,node), Pij.l(bus,node)*(-const2.m(bus)+
    const2.m(node)))/2 ;
display report,Pij.l,Congestioncost;
```

The total operating costs will be 1035 \$/h and the LMP for all buses are equal to 11 \$/h. This means that if the demand value at any bus increases for 1 MW then the operating cost will increase by 11 \$/h. Since the generating unit 1 is generating power at its maximum limit (because it is cheaper) the additional demand should be supplied by generating unit 2 with operating cost equal to 11 \$/MW h. The detailed three-bus optimal power flow solution with ($P_{ij}^{max} = 100$ MW) is given in Table 6.2.

Question: What would happen if the flow limit of the branch connecting bus 1 to bus 2 is reduced to 50 MW.

Answer: The answer is straightforward. The operating cost might increase but how much? We can easily decrease the flow limit in the GCode 6.2 (table Branch). The new operating cost would be 1056.250 \$/h. The LMP values are different for each bus in this case. The LMP values are $\lambda_1 = 10$ \$/MW h, $\lambda_2 = 11.625$ \$/MW h, $\lambda_3 = 11$ \$/MW h. The power flow solution is shown in Fig. 6.4. The LMP value in bus 1 is $\lambda_1 = 10$ \$/MW h because if the load increases in this node it will be supplied by generator 1 (which the operating costs are 10 \$/MW h). The LMP value in bus 3 is $\lambda_3 = 11$ \$/MW h because any increase in load in this node should be supplied by generator 2 (which the operating costs are 11 \$/MW h). Generator 1

Table 6.2 The three-bus optimal power flow solution ($P_{ij}^{max} = 100$ MW)

Bus (i)	P_g (MW)	δ_i (rad)	L_i (MW)	λ_i (\$/MW h)
1	65	0.02	0	11
2	0	−0.10	100	11
3	35	0.00	0	11



Fig. 6.4 The three-bus network with $P_{12}^{\max} = 50\,\text{MW}$

cannot send more power by line $1-2$ since it is congested and the direction of the power flow in branch $3-1$ is from bus 3 to bus 1. Finally, The LMP value in bus 2 is $\lambda_3 = 11.625$ \$/MW h because some of this demand will be supplied by generating unit 1 and some part will be supplied by unit 2. If the demand should pay the LMP value for every MW h then the total payment by the demand would be 100 MW $\times 11.625$ \$/MW h = 1162.5 \$/h. On the other hand, the generating units are also paid based on the LMP value of the connection point. The total payments to the generating units would be $43.75 \times 10 + 56.25 \times 11 = 1056.25$ \$/h. As it can be seen, there is a difference between what demand pays and what generating units receive. This surplus money is equal to $1162.5 - 1056.25 = 106.25$ \$/h. This is also called the congestion cost. Another technique for calculating the congestion costs (C_{cg}) is using the following formula [3]:

$$C_{cg} = \sum_{ij} P_{ji}(\lambda_i - \lambda_j) \tag{6.3}$$

The power flow solution of three-bus network with $P_{12}^{\max} = 50\,\text{MW}$ is shown in Fig. 6.4.

The detailed three-bus optimal power flow solution with ($P_{12}^{\max} = 50\,\text{MW}$) is given in Table 6.3.

Table 6.3 The three-bus optimal power flow solution ($P_{12}^{max} = 50\,MW$)

Bus (i)	P_g (MW)	δ_i (rad)	L_i (MW)	λ_i ($/MW h)
1	43.75	−0.025	0	10
2	0	−0.125	100	11.625
3	56.25	0	0	11



Fig. 6.5 The five-bus network data

6.1.2 Five-Bus Network DC-OPF

The five-bus network data is shown in Fig. 6.5. This example is taken from [4].

The GAMS code developed for solving OPF in five-bus network is provided in GCode 6.3.

GCode 6.3 The OPF GAMS code for five-bus network, Example (Sect. 6.1.2)

```
Sets bus  /1*5/, slack(bus) /1/  , Gen /g1*g5/;
Scalars Sbase /100/  ;  alias(bus,node);
Table GenData(Gen,*)
    b      pmin  pmax
g1  14     0     40
g2  15     0     170
g3  30     0     520
g4  40     0     200
g5  20     0     600  ;
set GBconect(bus,Gen) connectivity index of generating unit
/1        .      g1
 1        .      g2
 3        .      g3
 4        .      g4
 5        .      g5  /  ;
Table BusData(bus,*) Demands of each bus in MW
             Pd
```

```
2           300
3           300
4           400;
set conex             Bus connectivity matrix
/1    .    2
2    .    3
3    .    4
4    .    1
4    .    5
5    .    1/;
conex(bus,node)$conex(node,bus)=1;
Table branch(bus,node,*)
                 x          Limit
1    .    2    0.0281     400
1    .    4    0.0304     400
1    .    5    0.0064     400
2    .    3    0.0108     400
3    .    4    0.0297     400
4    .    5    0.0297     240 ;
branch(bus,node,'x')$(branch(bus,node,'x')=0)=branch(node,bus,'x'
    );
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node
    ,bus,'Limit');
branch(bus,node,'bij')$conex(bus,node) =1/branch(bus,node,'x');
Variables OF, Pij(bus,node),Pg(Gen),delta(bus);
Equations const1,const2,const3;
const1(bus,node)$conex(bus,node).. Pij(bus,node)=e=
 branch(bus,node,'bij')*(delta(bus)-delta(node));
const2(bus) .. +sum(Gen$GBconect(bus,Gen),Pg(Gen))-BusData(bus,'
    pd')/Sbase=e=
                               +sum(node$conex(node,bus),Pij(bus,
                                  node));
const3    .. OF=g=sum(Gen,Pg(Gen)*GenData(Gen,'b')*Sbase);
Model loadflow      /const1,const2,const3/;
Pg.lo(Gen)=GenData(Gen,'Pmin')/Sbase;
Pg.up(Gen)=GenData(Gen,'Pmax')/Sbase;
delta.up(bus)=pi;
delta.lo(bus)=-pi;
delta.fx(slack)=0;
Pij.up(bus,node)$((conex(bus,node)))= branch(bus,node,'Limit')/
    Sbase;
Pij.lo(bus,node)$((conex(bus,node)))=
-branch(bus,node,'Limit')/Sbase;
solve loadflow minimizing OF using lp;
parameter report(bus,*),Congestioncost;
report(bus,'Gen(MW)')= sum(Gen$GBconect(bus,Gen),Pg.l(Gen))*sbase
    ;
report(bus,'Angle')=delta.l(bus);
report(bus,'load(MW)')= BusData(bus,'pd');
report(bus,'LMP($/MWh)')=const2.m(bus)/sbase ;
Congestioncost= sum((bus,node),
Pij.l(bus,node)*(-const2.m(bus)+const2.m(node)))/2;
Display report,Pij.l,Congestioncost;
```



Fig. 6.6 The five-bus network flows and directions

Table 6.4 The optimal power flow solution in five-bus network

Bus (i)	P_g (MW)	δ_i (rad)	L_i (MW)	λ_i (\$/MW h)
1	210.000	0.000	0	16.977
2	0.000	−0.070	300	26.384
3	323.495	−0.065	300	30.000
4	0.000	−0.057	400	39.943
5	466.505	0.014	0	10.000

The minimum operating cost is 17,479.897 $/h. The network is highly congested, and the LMP values are different in various buses. The five-bus network flows and directions are depicted in Fig. 6.6. The detailed five-bus optimal power flow solution is given in Table 6.4.

6.1.3 IEEE Reliability Test System 24 Bus

The IEEE RTS 24-bus network is shown in Fig. 6.7. It is a transmission network with the voltage levels of 138 kV, 230 kV, and Sbase = 100 MVA. The branch data for IEEE RTS 24-bus network is given in Table 6.5 [5]. The from bus, to bus, reactance (X), resistance (r), total line charging susceptance (b), and MVA rating (MVA) are specified in this table. The parallel lines in MATPOWER are merged, and the resultants are given in Table 6.5. The generation data for IEEE RTS 24-bus network is given in Table 6.6. The data of generating units in this network is inspired by Conejo et al. [6] and Bouffard et al. [7] with some modifications. The slack bus is bus 13 in this network.



Fig. 6.7 IEEE RTS 24-bus network

Table 6.5 Branch data for IEEE RTS 24-bus network

From	To	*r*(pu)	x(pu)	*b*(pu)	Rating (MVA)	From	To	*r*(pu)	x(pu)	*b*(pu)	Rating (MVA)
1	2	0.0026	0.0139	0.4611	175	11	13	0.0061	0.0476	0.0999	500
1	3	0.0546	0.2112	0.0572	175	11	14	0.0054	0.0418	0.0879	500
1	5	0.0218	0.0845	0.0229	175	12	13	0.0061	0.0476	0.0999	500
2	4	0.0328	0.1267	0.0343	175	12	23	0.0124	0.0966	0.2030	500
2	6	0.0497	0.1920	0.0520	175	13	23	0.0111	0.0865	0.1818	500
3	9	0.0308	0.1190	0.0322	175	14	16	0.0050	0.0389	0.0818	500
3	24	0.0023	0.0839	0.0000	400	15	16	0.0022	0.0173	0.0364	500
4	9	0.0268	0.1037	0.0281	175	15	21	0.0032	0.0245	0.2060	1000
5	10	0.0228	0.0883	0.0239	175	15	24	0.0067	0.0519	0.1091	500
6	10	0.0139	0.0605	2.4590	175	16	17	0.0033	0.0259	0.0545	500
7	8	0.0159	0.0614	0.0166	175	16	19	0.0030	0.0231	0.0485	500
8	9	0.0427	0.1651	0.0447	175	17	18	0.0018	0.0144	0.0303	500
8	10	0.0427	0.1651	0.0447	175	17	22	0.0135	0.1053	0.2212	500
9	11	0.0023	0.0839	0.0000	400	18	21	0.0017	0.0130	0.1090	1000
9	12	0.0023	0.0839	0.0000	400	19	20	0.0026	0.0198	0.1666	1000
10	11	0.0023	0.0839	0.0000	400	20	23	0.0014	0.0108	0.0910	1000
10	12	0.0023	0.0839	0.0000	400	21	22	0.0087	0.0678	0.1424	500

Table 6.6 Generation data for IEEE RTS 24-bus network

Gen	Bus	$P_i^{\max}$	$P_i^{\min}$	b_i (\$/MW)	Cs_i (\$)	Cd_i (\$)	RU_i (MW h^{-1})	RD_i (MW h^{-1})	SU_i (MW h^{-1})	SD_i (MW h^{-1})	UT_i(h)	DT_i(h)	$u_{i,t=0}$	U_i^0 (h)	S_i^0 (h)
$g1$	18	400	100	5.47	0	0	47	47	105	108	1	1	1	5	0
$g2$	21	400	100	5.47	0	0	47	47	106	112	1	1	1	6	0
$g3$	1	152	30.4	13.32	1430.4	1430.4	14	14	43	45	8	4	1	2	0
$g4$	2	152	30.4	13.32	1430.4	1430.4	14	14	44	57	8	4	1	2	0
$g5$	15	155	54.25	16	0	0	21	21	65	77	8	8	0	0	2
$g6$	16	155	54.25	10.52	312	312	21	21	66	73	8	8	1	10	0
$g7$	23	310	108.5	10.52	624	624	21	21	112	125	8	8	1	10	0
$g8$	23	350	140	10.89	2298	2298	28	28	154	162	8	8	1	5	0
$g9$	7	350	75	20.7	1725	1725	49	49	77	80	8	8	0	0	2
$g10$	13	591	206.85	20.93	3056.7	3056.7	21	21	213	228	12	10	0	0	8
$g11$	15	60	12	26.11	437	437	7	7	19	31	4	2	0	0	1
$g12$	22	300	0	0	0	0	35	35	315	326	0	0	1	2	0

6.1.3.1 IEEE-RTS: Base Case

In this case, it is assumed that the network is intact and all branches and generating units are working in normal condition. The minimum operating cost is $29,574.275 obtained by using the GCode 6.4.

GCode 6.4 The OPF GAMS code for IEEE Reliability test 24-bus network, Example (Sect. 6.1.3)

```
sets bus  /1*24/,slack(bus) /13/,Gen /g1*g12/; scalar Sbase
    /100/; alias(bus,node);
Table GenData(Gen,*)  Generating units characteristics
     Pmax Pmin    b      CostsD  costst RU    RD    SU    SD    UT    DT
           uini U0    So
g1    400  100     5.47  0       0       47    47    105   108   1     1
      1     5     0    ;
set GBconect(bus,Gen) connectivity index of each generating unit
    to each bus
/18        .          g1  /  ;
Table BusData(bus,*) Demands of each bus in MW/MVar
      Pd    Qd
1     108   22;
Table branch(bus,node,*)     Network technical characteristics
                    r        x        b       limit
1      .     2      0.0026   0.0139   0.4611  175;
parameter conex(bus,node);
conex(bus,node)$branch(bus,node,'limit')=1;
conex(bus,node)$(conex(node,bus))=1;
branch(bus,node,'x')$(branch(bus,node,'x')=0)=branch(node,bus,'x'
    );
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node
    ,bus,'Limit');
branch(bus,node,'bij')$conex(bus,node) =1/branch(bus,node,'x');
Variables OF, Pij(bus,node),Pg(Gen),delta(bus);
Equations const1,const2,const3;
const1(bus,node)$( conex(bus,node)) ..
Pij(bus,node)=e= branch(bus,node,'bij')*(delta(bus)-delta(node));
const2(bus) .. +sum(Gen$GBconect(bus,Gen),Pg(Gen))-BusData(bus,'
    pd')/Sbase=e=
+sum(node$conex(node,bus),Pij(bus,node));
const3      .. OF=g=sum(Gen,Pg(Gen)*GenData(Gen,'b')*Sbase);
Model loadflow      /const1,const2,const3/;
Pg.lo(Gen)=GenData(Gen,'Pmin')/Sbase;Pg.up(Gen)=GenData(Gen,
'Pmax')/Sbase;
delta.up(bus)=pi/2; delta.lo(bus)=-pi/2;delta.fx(slack)=0;
Pij.up(bus,node)$((conex(bus,node)))=1* branch(bus,node,'Limit')/
    Sbase;
Pij.lo(bus,node)$((conex(bus,node)))=-1*branch(bus,node,'Limit')/
    Sbase;
Solve loadflow minimizing OF using lp;
```

Table 6.7 Base case solution of IEEE RTS 24-bus network (branch flow limits are unchanged)

Bus	P_g (MW)	δ_i (rad)	Load (MW)	λ_i ($/MW h)
1	152	−0.150	108	20.7
2	152	−0.151	97	20.7
3		−0.113	180	20.7
4		−0.185	74	20.7
5		−0.191	71	20.7
6		−0.230	136	20.7
7	257.15	−0.105	125	20.7
8		−0.186	171	20.7
9		−0.136	175	20.7
10		−0.172	195	20.7
11		−0.044		20.7
12		−0.031		20.7
13	206.85	0.000	265	20.7
14		0.027	194	20.7
15	167	0.182	317	20.7
16	155	0.168	100	20.7
17		0.251		20.7
18	400	0.276	333	20.7
19		0.146	181	20.7
20		0.164	128	20.7
21	400	0.291		20.7
22	300	0.399		20.7
23	660	0.187		20.7
24		0.069		20.7

The base case solution of IEEE RTS 24-bus network without changing the branch flow limits is described in Table 6.7. As it can be observed in Table 6.7, the LMP values are all the same and equal to 20.7 $/MW h. This is because there is no congestion in this network for the given loading values. The congestion cost would be zero in this case.

6.1.3.2 IEEE-RTS: Branch Flow Limit Reduction

Now the branch flow limits are reduced by 30%. The problem is solved again, and the angle values are found as given in Table 6.8. The minimum operating cost is $29,747.745 obtained by using the GCode 6.4. The congestion cost is $4597.217 in this case. The solution of IEEE RTS 24-bus network (branch flow limits are reduced by 30%) is given in Table 6.8. Reducing the branch flow limit will not only increase the operating cost but also makes the LMP values different across the network.

Table 6.8 Solution of IEEE RTS 24-bus network (branch flow limits are reduced by 30%)

Bus	P_g (MW)	δ_i (rad)	Load (MW)	λ_i ($/MW h)
1	152.00	−0.162	108	20.66
2	152.00	−0.163	97	20.71
3		−0.129	180	19.11
4		−0.197	74	20.85
5		−0.203	71	20.98
6		−0.241	136	21.17
7	247.50	−0.130	125	20.70
8		−0.205	171	21.14
9		−0.147	175	20.96
10		−0.183	195	21.32
11		−0.053		22.69
12		−0.038		20.55
13	251.24	0.000	265	20.93
14		0.012	194	25.78
15	132.26	0.158	317	16.00
16	155.00	0.149	100	15.68
17		0.230		15.79
18	400.00	0.255	333	15.85
19		0.131	181	16.86
20		0.152	128	17.87
21	400.00	0.269		15.89
22	300.00	0.378		15.85
23	660.00	0.177		18.42
24		0.049		17.19

6.1.3.3 IEEE-RTS: Branch Outage

In this case, some branch contingencies are examined. In order to simulate the branch outage, the following equation should be satisfied.

$$P_{ij} - \frac{\delta_i - \delta_j}{x_{ij}} \leq M\xi_{ij} \tag{6.4}$$

$$P_{ij} - \frac{\delta_i - \delta_j}{x_{ij}} \geq -M\xi_{ij} \tag{6.5}$$

where ξ_{ij} is a binary parameter which shows the status of the branch connecting bus i to bus j. In the developed GAMS code, if the branch limit is set to 0 it is considered as an outaged branch. This is because $const1$ which models the flow calculation in line ij is calculated for every branch which has $conex(bus, node) > 0$. Now some different contingencies are evaluated as follows:

- Contingency 1: branch $\ell_{20-19}, \ell_{12-23}$ are out. Congestion costs are \$4905.000 and OF = \$29,888.196. The congested lines are ℓ_{13-23}. The LMP values are not the same in different buses.
- Contingency 2: branch $\ell_{14-16}, \ell_{16-19}$ are out. Congestion costs are \$6224.250 and OF = \$32,199.855. The congested lines are ℓ_{24-3}, ℓ_{8-7}. The LMP values are not the same in different buses.
- Contingency 3: branch ℓ_{1-5}, ℓ_{4-2} are out. Congestion costs are \$0 and OF = \$29,574.275. No line would be congested and therefore the LMP values are the same in all buses.

The OPF solutions for these three contingency cases are provided in Table 6.9.

Table 6.9 Solution of IEEE RTS 24-bus network (branch outage contingencies)

	Contingency 1			Contingency 2			Contingency 3		
Bus	P_g (MW)	δ_i (rad)	λ_i (\$/MW h)	P_g (MW)	δ_i (rad)	λ_i (\$/MW h)	P_g (MW)	δ_i (rad)	λ_i (\$/MW h)
1	152	−0.218	20.70	152	−0.106	20.93	152	−0.050	20.70
2	152	−0.219	20.70	152	−0.112	20.93	152	−0.054	20.70
3		−0.193	20.70		0.055	20.93		−0.088	20.70
4		− 0.252	20.70		− 0.156	20.93		−0.215	20.70
5		− 0.257	20.70		-0.173	20.93		−0.239	20.70
6		− 0.294	20.70		− 0.227	20.93		−0.210	20.70
7	289.15	−0.123	20.70	300	−0.038	20.70	257.15	−0.108	20.70
8		−0.224	20.70		− 0.145	20.93		−0.189	20.70
9		−0.202	20.70		−0.116	20.93		−0.138	20.70
10		−0.235	20.70		−0.181	20.93		−0.176	20.70
11		−0.094	20.70		−0.122	20.93		−0.044	20.70
12		−0.116	20.70		−0.037	20.93		−0.032	20.70
13	206.85		20.70	436		20.93	206.85		20.70
14		−0.053	20.70		−0.203	20.93		0.029	20.70
15	167	0.077	20.70	66.25	0.598	5.47	167	0.189	20.70
16	155	0.060	20.70	54.25	0.631	5.47	155	0.173	20.70
17		0.144	20.70		0.691	5.47		0.257	20.70
18	400	0.171	20.70	400	0.706	5.47	400	0.283	20.70
19		0.019	20.70		0.073	20.93		0.150	20.70
20		0.419	10.89		0.109	20.93		0.167	20.70
21	400	0.185	20.70	329.5	0.711	5.47	400	0.298	20.70
22	300	0.293	20.70	300	0.827	5.47	300	0.405	20.70
23	628	0.433	10.89	660	0.142	20.93	660	0.190	20.70
24		−0.026	20.70		0.390	5.47		0.083	20.70

6.1.3.4 IEEE-RTS: Generator Outage

In this case, some generating units are out of service. This can be because of unplanned outage or maintenance purpose. Now some different contingencies are evaluated as follows:

- Contingency 1: The generating unit g_9 (connected to bus 7) is out. Solving the problem shows that no line is congested but removing the g_9 will cause an increase in total operating costs which becomes OF = \$29,633.420.
- Contingency 2: The generating unit g_5 (connected to bus 15) is out. Solving the problem shows that line ℓ_{8-7} is congested but removing the g_5 will cause an increase in total operating costs which becomes OF = \$30,328.570. The LMP values are different on different buses, and the congestion costs are \$40.250.
- Contingency 3: The generating unit g_8 (connected to bus 23) is out. Solving the problem shows that line ℓ_{8-7} is congested but removing the g_8 will cause an increase in total operating costs which becomes OF = \$33,078.420. The LMP values are different on different buses, and the congestion costs are \$40.250.
- Contingency 4: The generating unit g_{10} (connected to bus 13) is out. The problem becomes infeasible, and GAMS cannot find any solution for it. This is because no solution is found which satisfies the technical constraints. Some load shedding action should take place.

The OPF solutions for first three contingency cases are given in Table 6.10. In order to analyze the contingency case 4, the load shedding is modeled using a virtual generating unit connected to bus i as shown in Fig. 6.8. The production cost of this specific unit is set to a high value. It is usually called the value of the loss of load. The production level of this generating unit is equal to the load shedding that should happen in bus i. The maximum power of this unit is equal to the load connected to the bus i ($0 \leq \text{LS}_i \leq L_i$). The DC-OPF problem considering the load shedding concept is formulated in (6.6).

$$\text{OF} = \sum_{g \in \Omega_G} a_g (P_g)^2 + b_g P_g + c_g + \sum_i \text{VOLL} \times \text{LS}_i \tag{6.6a}$$

$$P_{ij} = \frac{\delta_i - \delta_j}{x_{ij}} \quad ij \in \Omega_\ell \tag{6.6b}$$

$$\sum_{g \in \Omega_G^i} P_g + \text{LS}_i - L_i = \sum_{j \in \Omega_\ell^i} P_{ij} : \lambda_i \quad i \in \Omega_B \tag{6.6c}$$

$$-P_{ij}^{\max} \leq P_{ij} \leq P_{ij}^{\max} \quad ij \in \Omega_\ell \tag{6.6d}$$

$$P_g^{\min} \leq P_g \leq P_g^{\max} \tag{6.6e}$$

$$0 \leq \text{LS}_i \leq L_i \tag{6.6f}$$

Table 6.10 Solution of IEEE RTS 24-bus network (generator outage contingencies)

	Contingency 1			Contingency 2			Contingency 3		
Bus	P_g (MW)	δ_i (rad)	λ_i (\$/MW h)	P_g (MW)	δ_i (rad)	λ_i (\$/MW h)	P_g (MW)	δ_i (rad)	λ_i (\$/MW h)
1	152	−0.246	20.93	152	−0.171	20.93	152	−0.186	20.93
2	152	−0.248	20.93	152	−0.172	20.93	152	−0.187	20.93
3		−0.195	20.93		−0.156	20.93		−0.165	20.93
4		−0.283	20.93		−0.204	20.93		−0.219	20.93
5		−0.291	20.93		−0.208	20.93		−0.224	20.93
6		−0.331	20.93		−0.244	20.93		−0.261	20.93
7		−0.576	20.93	300	−0.058	20.7	300	−0.075	20.7
8		−0.499	20.93		−0.165	20.93		−0.182	20.93
9		−0.235	20.93		−0.153	20.93		−0.169	20.93
10		−0.275	20.93		−0.184	20.93		−0.202	20.93
11		−0.097	20.93		−0.065	20.93		−0.076	20.93
12		−0.082	20.93		−0.044	20.93		−0.071	20.93
13	464	0.000	20.93	319	0.000	20.93	514	0.000	20.93
14		−0.024	20.93		−0.020	20.93		−0.034	20.93
15	167	0.129	20.93	12	0.095	20.93	167	0.098	20.93
16	155	0.118	20.93	155	0.099	20.93	155	0.081	20.93
17		0.200	20.93		0.175	20.93		0.165	20.93
18	400	0.226	20.93	400	0.198	20.93	400	0.191	20.93
19		0.103	20.93		0.092	20.93		0.038	20.93
20		0.126	20.93		0.122	20.93		0.038	20.93
21	400	0.240	20.93	400	0.210	20.93	400	0.206	20.93
22	300	0.348	20.93	300	0.320	20.93	300	0.314	20.93
23	660	0.152	20.93	660	0.152	20.93	310	0.051	20.93
24		0.006	20.93		−0.001	20.93		−0.003	20.93



Fig. 6.8 Load shedding modeling using virtual generating unit

The GAMS code for modeling and calculating the load shedding values is given in GCode 6.5:

GCode 6.5 The OPF GAMS code for IEEE Reliability test 24-bus network with load shedding modeling

```
Sets
bus   /1*24/
slack(bus) /13/
Gen /g1*g12/
scalars
Sbase /100/
VOLL /10000/;
alias(bus,node);
table GenData(Gen,*)  Generating units characteristics
      Pmax Pmin    b
*******REMOVED FOR SAVING SPACE******
;
set GBconect(bus,Gen) connectivity index of each generating unit
     to each bus
/
**REMOVED FOR SAVING SPACE******
****************************** / ;
Table BusData(bus,*) Demands of each bus in MW
       Pd    Qd
*******REMOVED FOR SAVING SPACE******
;

table branch(bus,node,*)     Network technical characteristics
                   r         x         b        limit
*******REMOVED FOR SAVING SPACE******
;
branch(bus,node,'x')$(branch(bus,node,'x')=0)=branch(node,bus,'x'
     );
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node
     ,bus,'Limit');
branch(bus,node,'bij')$branch(bus,node,'Limit') =1/branch(bus,
     node,'x');
parameter conex(bus,node);
conex(bus,node)$(branch(bus,node,'limit')and branch(node,bus,'
     limit'))=1;
conex(bus,node)$(conex(node,bus))=1;
Variables
OF,Pij(bus,node),Pg(Gen),delta(bus),lsh(bus);
Equations
const1,const2,const3;
const1(bus,node)$( conex(bus,node)) .. Pij(bus,node)=e=
                          branch(bus,node,'bij')*(delta(bus)-delta(node
                              ));
const2(bus) ..lsh(bus) +sum(Gen$GBconect(bus,Gen),Pg(Gen))-
     BusData(bus,'pd')/Sbase
                                                =e=+sum(node$conex(node,bus
                                                     ),Pij(bus,node));
```

```
const3      .. OF=g=sum((bus,Gen)$GBconect(bus,Gen),Pg(Gen)*GenData
     (Gen,'b')*Sbase)+sum(bus,VOLL*lsh(bus)*Sbase);
model loadflow      /const1,const2,const3/;
Pg.lo(Gen)=GenData(Gen,'Pmin')/Sbase;
Pg.up(Gen)=GenData(Gen,'Pmax')/Sbase;
delta.up(bus)=pi/2;
delta.lo(bus)=-pi/2;
delta.fx(slack)=0;
Pij.up(bus,node)$((conex(bus,node)))=1* branch(bus,node,'Limit')/
     Sbase;
Pij.lo(bus,node)$((conex(bus,node)))=-1*branch(bus,node,'Limit')/
     Sbase;
lsh.up(bus)= BusData(bus,'pd')/Sbase;
lsh.lo(bus)= 0;
solve loadflow minimizing OF using lp;
parameter report(bus,*),Congestioncost;
report(bus,'Gen(MW)')= 1*sum(Gen$GBconect(bus,Gen),Pg.l(Gen))*
     sbase;
report(bus,'Angle')=delta.l(bus);
report(bus,'load(MW)')= BusData(bus,'pd');
report(bus,'LSH')=lsh.l(bus)*sbase ;
report(bus,'LMP($/MWh)')=const2.m(bus)/sbase ;
Congestioncost= sum((bus,node), Pij.l(bus,node)*(-const2.m(bus)+
     const2.m(node)))/2 ;
display report,Pij.l,Congestioncost;
execute_unload "opf.gdx" report
execute 'gdxxrw.exe  OPF.gdx o=OPF.xls par=report rng=report!A1'
display GBconect;
```

The OPF solution for contingency case 4 is given in Table 6.11. The simulation results show that the demand in bus 18 should be reduced by 116 MW as given in Table 6.11. It should be noted that the LMP values are not meaningful in this case. The LMP values are calculated and shown in Table 6.12.

In order to calculate the LMP in a network with load shedding, the calculated amount of load shedding should be curtailed from the demand in the specific bus (here bus 18) and the problem should be resolved.

Nomenclature

Indices and Sets

g	Index of thermal generating units.
i, j	Index of network buses.
Ω_G	Set of all thermal generating units.
Ω_G^i	Set of all thermal generating units connected to bus i.
Ω_ℓ^i	Set of all buses connected to bus i.
Ω_B	Set of network buses.
Ω_ℓ	Set of network branches.

Table 6.11 Solution of IEEE RTS 24-bus network (generator g_{10} outage with load shedding modeling)

Bus	P_g (MW)	δ_i (rad)	L_i (MW)	LSH_i (MW)
1	152	−0.095	108	
2	152	−0.097	97	
3		−0.042	180	
4		−0.132	74	
5		−0.140	71	
6		−0.181	136	
7	300	0.006	125	
8		−0.101	171	
9		−0.084	175	
10		−0.124	195	
11		−0.004		
12		−0.001		
13		0.000	265	
14		0.093	194	
15	215	0.285	317	
16	155	0.259	100	
17		0.360		
18	400	0.396	333	116
19		0.220	181	
20		0.222	128	
21	400	0.405		
22	300	0.511		
23	660	0.237		
24		0.160		

Parameters

L_i Electric power demand in bus i
a_g, b_g, c_g Fuel cost coefficients of thermal unit g.
$P_g^{\max/\min}$ Maximum/minimum limits of power generation of thermal unit g.
$P_{ij}^{\max}$ Maximum power flow limits of branch connecting bus i to j.
x_{ij} Reactance of branch connecting bus i to j.
r_{ij} Resistance of branch connecting bus i to j.

Variables

P_{ij} Active power flow of branch connecting bus i to j.
P_g Active power generated by thermal unit g(MW).
λ_i Locational marginal price in bus i ($/MW h).
LS_i Load shedding in bus i (MW).
OF Total operating costs ($/h).
δ_i Voltage angle in bus i (rad).

Table 6.12 Solution of IEEE RTS 24-bus network (generator g_{10} outage with load shedding modeling and LMP calculation)

Bus	P_g (MW)	δ_i (rad)	L_i (MW)	λ_i (\$/MW h)
1	152	−0.095	108	26.110
2	152	−0.097	97	26.110
3		−0.042	180	26.110
4		−0.132	74	26.110
5		−0.140	71	26.110
6		−0.181	136	26.110
7	300	0.006	125	20.700
8		−0.101	171	26.110
9		−0.084	175	26.110
10		−0.124	195	26.110
11		−0.004		26.110
12		−0.001		26.110
13		0.000	265	26.110
14		0.093	194	26.110
15	215	0.285	317	26.110
16	155	0.259	100	26.110
17		0.360		26.110
18	400	0.396	217	26.110
19		0.220	181	26.110
20		0.222	128	26.110
21	400	0.405		26.110
22	300	0.511		26.110
23	660	0.237		26.110
24		0.160		26.110

6.2 Multi-Period Optimal Wind-DC OPF

The load flow model considering wind power and load shedding is shown in Fig. 6.9. The multi-period wind DC-OPF is formulated as follows:

$$\text{OF} = \sum_{g,t} a_g (P_{g,t})^2 + b_g P_{g,t} + c_g + \sum_{i,t} \text{VOLL} \times \text{LS}_{i,t} + \text{VWC} \times P^{\text{wc}}_{i,t} \tag{6.7a}$$

$$\sum_{g \in \Omega^i_G} P_{g,t} + \text{LS}_{i,t} + P^w_{i,t} - L_{i,t} = \sum_{j \in \Omega^i_\ell} P_{ij,t} : \lambda_{i,t} \tag{6.7b}$$

$$P_{ij,t} = \frac{\delta_{i,t} - \delta_{j,t}}{x_{ij}} \tag{6.7c}$$

$$-P^{\max}_{ij} \le P_{ij,t} \le P^{\max}_{ij} \tag{6.7d}$$

$$P^{\min}_g \le P_{g,t} \le P^{\max}_g \tag{6.7e}$$



Fig. 6.9 Load flow model considering wind power and load shedding

$$P_{g,t} - P_{g,t-1} \leq \mathrm{RU}_g \tag{6.7f}$$

$$P_{g,t-1} - P_{g,t} \leq \mathrm{RD}_g \tag{6.7g}$$

$$0 \leq \mathrm{LS}_{i,t} \leq L_{i,t} \tag{6.7h}$$

$$P_{i,t}^{\mathrm{wc}} = w_t \Lambda_i^w - P_{i,t}^w \tag{6.7i}$$

$$0 \leq P_{i,t}^w \leq w_t \Lambda_i^w \tag{6.7j}$$

The objective function in (6.7a) consists of operating costs of thermal units, load shedding costs, and wind curtailment costs. The nodal power balance is formulated in (6.7b). This equation is also providing the LMP value in bus i at time t ($\lambda_{i,t}$). The active flow in branch connecting bus i to bus j is calculated in (6.7c). The branch flow limits of every branch is formulated in (6.7d). The operating limits of the thermal generating unit are modeled in (6.7e). The ramp rates of thermal units are described in (6.7f) and (6.7g). The load shedding of bus i is limited to the existing demand at time t as stated in (6.7h). The wind power curtailment is formulated in (6.7i). At any bus i hosting wind turbine, the amount of wind power generation depends on wind power availability and wind power capacity as described in (6.7j). It should bear in mind that the demand and wind power availability change vs time. These variations are reflected in $L_{i,t}$ and w_t, respectively. The wind power connection points to the network are indicated in Fig. 6.10.

The wind-demand variation pattern vs time is shown in Fig. 6.11. The optimal thermal unit power generation schedules of multi-period wind DC-OPF is given in Table 6.13.



Fig. 6.10 Wind power connection points to the network



Fig. 6.11 Wind-demand variation patterns vs time

Table 6.13 Thermal unit power generation schedules of multi-period wind DC-OPF

Thermal unit	Bus	t_1	t_2	t_3	t_4	t_5	t_6	t_7	t_8	t_9	t_{10}	t_{11}	t_{12}	t_{13}	t_{14}	t_{15}	t_{16}	t_{17}	t_{18}	t_{19}	t_{20}	t_{21}	t_{22}	t_{23}	t_{24}
$g1$	18.0	4.0	4.0	3.5	3.1	2.7	2.3	2.6	3.1	3.5	4.0	4.0	4.0	4.0	3.6	3.1	3.1	3.5	4.0	4.0	4.0	4.0	3.5	3.1	2.6
$g2$	21.0	4.0	3.8	3.3	2.8	2.4	2.1	2.6	3.1	3.5	4.0	4.0	4.0	4.0	3.8	3.5	3.1	3.5	4.0	4.0	4.0	4.0	3.5	3.1	3.0
$g3$	1.0	0.4	0.3	0.3	0.3	0.3	0.3	0.3	0.3	0.3	0.4	0.4	0.4	0.4	0.3	0.3	0.4	0.6	0.7	0.9	0.7	0.6	0.4	0.3	0.3
$g4$	2.0	0.4	0.3	0.3	0.3	0.3	0.3	0.3	0.3	0.3	0.4	0.4	0.3	0.3	0.3	0.3	0.4	0.6	0.7	0.7	0.7	0.6	0.4	0.3	0.3
$g5$	15.0	0.5	0.5	0.5	0.5	0.5	0.5	0.5	0.5	0.5	0.6	0.5	0.5	0.5	0.5	0.5	0.5	0.8	1.0	1.0	1.2	1.0	0.8	0.5	0.5
$g6$	16.0	0.8	0.5	0.5	0.5	0.5	0.5	0.5	0.5	0.8	1.0	1.2	1.4	1.2	1.0	0.9	1.1	1.3	1.6	1.6	1.4	1.2	1.0	0.8	0.5
$g7$	23.0	1.4	1.2	1.1	1.1	1.1	1.1	1.1	1.2	1.4	1.6	1.8	2.0	2.2	2.1	2.3	2.5	2.7	2.9	3.1	2.9	2.7	2.5	2.3	2.1
$g8$	23.0	1.7	1.4	1.4	1.4	1.4	1.4	1.4	1.4	1.7	2.0	2.2	2.5	2.3	2.1	2.3	2.6	2.9	3.2	3.3	3.0	2.7	2.4	2.1	1.9
$g9$	7.0	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	0.8	1.2	1.7	1.2	1.1	0.9	0.8	0.8	0.8
$g10$	13.0	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.1	2.3	2.5	2.3	2.1	2.1	2.1	2.1	2.1
$g11$	15.0	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.1	0.2	0.3	0.2	0.1	0.1	0.1	0.1	0.1
$g12$	22.0	3.0	3.0	3.0	2.9	3.0	3.0	2.7	2.7	3.0	3.0	3.0	3.0	3.0	3.0	2.7	2.3	2.7	3.0	3.0	3.0	3.0	3.0	2.7	3.0

GCode 6.6 Multi-period DC-OPF for modified IEEE RTS 24-bus system

```
Sets bus /1*24/ ,slack(bus) /13/,Gen /g1*g12/, t /t1*t24/;
Scalars Sbase /100/ ,VOLL /10000/,VWC /50/; alias(bus,node);
table GD(Gen,*) Generating units characteristics
     Pmax Pmin  b     CostsD costst RU   RD   SU   SD   UT   DT
           uini U0   So
* Removed for saving space
Set GB(bus,Gen) connectivity index of each generating unit to
    each bus
/* Removed for saving space / ;
Table BusData(bus,*) Demands of each bus in MW
**********Omitted for saving space ;
table branch(bus,node,*)   Network technical characteristics
**********Omitted for saving space ;
Table WD(t,*)
       w                  d
t1     0.0786666666666667 0.684511335492475
t2     0.0866666666666667 0.644122690036197
t3     0.117333333333333  0.61306915602972
t4     0.258666666666667  0.599733282530006
t5     0.361333333333333  0.588874071251667
t10    0.548              0.787007048961707
t20    0.561333333333333  0.936368832158506
t21    0.565333333333333  0.887597637645266
t22    0.556              0.809297008954087
t23    0.724              0.74585635359116
t24    0.84               0.733473042484283;
Parameter Wcap(bus)
/8    200
19   150
21   100/;
branch(bus,node,'x')$(branch(bus,node,'x')=0)=branch(node,bus,'x'
    );
branch(bus,node,'Limit')$(branch(bus,node,'Limit')=0)=branch(node
    ,bus,'Limit');
branch(bus,node,'bij')$branch(bus,node,'Limit') =1/branch(bus,
    node,'x');
branch(bus,node,'z')$branch(bus,node,'Limit')=sqrt(power(branch
(bus,node,'x'),2)+power(branch(bus,node,'r'),2));
branch(node,bus,'z')=branch(bus,node,'z');
parameter conex(bus,node);
conex(bus,node)$(branch(bus,node,'limit')and branch(node,bus,'
    limit'))=1;
conex(bus,node)$(conex(node,bus))=1;
Variables
OF,Pij(bus,node,t),Pg(Gen,t),delta(bus,t),lsh(bus,t),Pw(bus,t),pc
    (bus,t);
Equations
const1,const2,const3,const4,const5,const6;
const1(bus,node,t)$( conex(bus,node)) .. Pij(bus,node,t)=e=
branch(bus,node,'bij')*(delta(bus,t)-delta(node,t));
const2(bus,t) .. lsh(bus,t)$BusData(bus,'pd')
+Pw(bus,t)$Wcap(bus)+sum(Gen$GB(bus,Gen),Pg(Gen,t))
```

```
−WD(t,'d')*BusData(bus,'pd')/Sbase=e=+sum(node$conex(node,bus),
      Pij(bus,node,t));
const3      .. OF=g=sum((bus,Gen,t)$GB(bus,Gen),Pg(Gen,t)*GD(Gen,'b
      ')*Sbase)
                    +sum((bus,t),VOLL*lsh(bus,t)*Sbase$BusData
                    +VWC*Pc(bus,t)*sbase$Wcap(bus) );
const4(gen,t)  ..  pg(gen,t+1)−pg(gen,t)=l=GD(gen,'RU')/Sbase;
const5(gen,t)  ..  pg(gen,t−1)−pg(gen,t)=l=GD(gen,'RD')/Sbase;
const6(bus,t)$Wcap(bus)  ..  pc(bus,t)=e=WD(t,'w')*Wcap(bus)/Sbase−
      pw(bus,t);
model loadflow      /const1,const2,const3,const4,const5,const6/;
Pg.lo(Gen,t)=GD(Gen,'Pmin')/Sbase;  Pg.up(Gen,t)=GD(Gen,'Pmax')/
      Sbase;
delta.up(bus,t)=pi/2;  delta.lo(bus,t)=−pi/2;  delta.fx(slack,t)=0;
Pij.up(bus,node,t)$((conex(bus,node)))=1* branch(bus,node,'Limit'
      )/Sbase;
Pij.lo(bus,node,t)$((conex(bus,node)))=−1*branch(bus,node,'Limit'
      )/Sbase;
lsh.up(bus,t)= WD(t,'d')*BusData(bus,'pd')/Sbase;  lsh.lo(bus,t)=
      0;
Pw.up(bus,t)=WD(t,'w')*Wcap(bus)/Sbase;  Pw.lo(bus,t)=0;
Pc.up(bus,t)=WD(t,'w')*Wcap(bus)/Sbase;  Pc.lo(bus,t)=0;
Solve loadflow minimizing OF using lp;
```

The wind power generations at efferent buses vs time are shown in Fig. 6.12.



Fig. 6.12 Wind power generations at efferent buses vs time

The wind power curtailment is occurring in bus 19 at time t_{17} equal to 38.76 and also in bus 21 at time t_{16} equal to 15.32 MW and also at time t_{17} equal to 86.93 MW. The total operating cost (including the wind curtailment cost) would be $\$4.3229 \times 10^5$.

Nomenclature

Indices and Sets

g Index of thermal generating units.
w Index of wind turbine units.
i, j Index of network buses.
Ω_G Set of all thermal generating units.
Ω_G^i Set of all thermal generating units connected to bus i.
Ω_ℓ^i Set of all buses connected to bus i.
Ω_B Set of network buses.
Ω_ℓ Set of network branches.

Parameters

$w_{i,t}$ Availability of wind turbine connected to bus i at time t
Λ_i^w Capacity of wind turbine connected to bus i
$L_{i,t}$ Electric power demand in bus i at time t
a_g, b_g, c_g Fuel cost coefficients of thermal unit g.
$P_g^{\max/\min}$ Maximum/minimum limits of power generation of thermal unit g.
$P_{ij}^{\max}$ Maximum power flow limits of branch connecting bus i to j.
x_{ij} Reactance of branch connecting bus i to j.
r_{ij} Resistance of branch connecting bus i to j.
VOLL Value of loss of load (\$/MW h)
VWC Value of loss of wind (\$/MW h)

Variables

$P_{ij,t}$ Active power flow of branch connecting bus i to j at time t (MW).
$P_{g,t}$ Active power generated by thermal unit g at time t (MW) .
$P_{i,t}^w$ Active power generated by wind turbine connected to bus i at time t (MW).
$P_{i,t}^{wc}$ Curtailed power of wind turbine connected to bus i at time t (MW).
$\lambda_{i,t}$ Locational marginal price in bus i at time t (\$/MW h).
$\mathrm{LS}_{i,t}$ Load shedding in bus i at time t (MW).
OF Total operating costs (\$).
$\delta_{i,t}$ Voltage angle of bus i at time t (rad).

6.3 Multi-Period Optimal AC Power Flow

The AC power flow considering wind generation and load shedding is shown in Fig. 6.13.

The optimal power flow equations are described in (6.8) as follows:

$$\mathrm{OF} = \sum_{i,t} a_g (P^g_{i,t})^2 + b_g P^g_{i,t} + c_g + \sum_{i,t} \mathrm{VOLL} \times P^{\mathrm{LS}}_{i,t} + \mathrm{VWC} \times P^{\mathrm{wc}}_{i,t} \tag{6.8a}$$

$$P^g_{i,t} + P^{\mathrm{LS}}_{i,t} + P^w_{i,t} - P^L_{i,t} = \sum_{j \in \Omega^i_\ell} P_{ij,t} : \lambda^p_{i,t} \tag{6.8b}$$

$$Q^g_{i,t} + Q^{\mathrm{LS}}_{i,t} + Q^w_{i,t} - Q^L_{i,t} = \sum_{j \in \Omega^i_\ell} Q_{ij,t} : \lambda^q_{i,t} \tag{6.8c}$$

$$I_{ij,t} = \frac{V_{i,t}\angle\delta_{i,t} - V_{j,t}\angle\delta_{j,t}}{Z_{ij}\angle\theta_{ij}} + \frac{bV_{i,t}}{2}\angle\left(\delta_{i,t} + \frac{\pi}{2}\right) \tag{6.8d}$$

$$S_{ij,t} = (V_{i,t}\angle\delta_{i,t}) I^*_{ij,t} \tag{6.8e}$$

$$P_{ij,t} = \mathrm{real}\left\{S_{ij,t}\right\} = \frac{V^2_{i,t}}{Z_{ij}}\cos(\theta_{ij}) - \frac{V_{i,t}V_{j,t}}{Z_{ij}}\cos(\delta_{i,t} - \delta_{j,t} + \theta_{ij}) \tag{6.8f}$$

$$Q_{ij,t} = \mathrm{Img}\left\{S_{ij,t}\right\} = \frac{V^2_{i,t}}{Z_{ij}}\sin(\theta_{ij}) - \frac{V_{i,t}V_{j,t}}{Z_{ij}}\sin(\delta_{i,t} - \delta_{j,t} + \theta_{ij}) - \frac{bV^2_{i,t}}{2} \tag{6.8g}$$

$$-S^{\max}_{ij} \le S_{ij,t} \le S^{\max}_{ij} \tag{6.8h}$$

$$P^{g,\min}_i \le P^g_{i,t} \le P^{g,\max}_i \tag{6.8i}$$

$$Q^{g,\min}_i \le Q^g_{i,t} \le Q^{g,\max}_i \tag{6.8j}$$



Fig. 6.13 AC power flow considering wind generation and load shedding

$$P^g_{i,t} - P^g_{i,t-1} \leq \mathrm{RU}_g \tag{6.8k}$$

$$P^g_{i,t-1} - P^g_{i,t} \leq \mathrm{RD}_g \tag{6.8l}$$

$$0 \leq P^{\mathrm{LS}}_{i,t} \leq P^L_{i,t} \tag{6.8m}$$

$$0 \leq Q^{\mathrm{LS}}_{i,t} \leq Q^L_{i,t} \tag{6.8n}$$

$$P^{\mathrm{wc}}_{i,t} = w_t \Lambda^w_i - P^w_{i,t} \tag{6.8o}$$

$$0 \leq P^w_{i,t} \leq w_t \Lambda^w_i \tag{6.8p}$$

The GAMS code developed for solving the minimum cost OPF is given in GCode 6.7. The optimal active/reactive power generation of thermal units in MP-AC OPF are given in Figs. 6.14 and 6.15, respectively.



Fig. 6.14 Active power generation of thermal units in MP-AC OPF



Fig. 6.15 Reactive power generation of thermal units in MP-AC OPF

GCode 6.7 Multi-period AC OPF for modified IEEE RTS 24 bus system

```
Sets i  network buses /1*24/ ,slack(i) /13/, t /t1*t24/
GB(i) generating buses /1,2,7,15*16,18,21*23/;
scalars Sbase /100/  ,VOLL /10000/,VWC /50/; alias(i,j);
Table GenD(i,*)  Generating units characteristics
      pmax pmin   b      Qmax Qmin Vg    RU   RD
1     152  30.4   13.32  192  -50  1.035 21   21
2     152  30.4   13.32  192  -50  1.035 21   21
7     350  75     20.7   300  0    1.025 43   43
13    591  206.85 20.93  591  0    1.02  31   31
15    215  66.25  21     215  -100 1.014 31   31
16    155  54.25  10.52  155  -50  1.017 31   31
18    400  100    5.47   400  -50  1.05  70   70
21    400  100    5.47   400  -50  1.05  70   70
22    300  0      0      300  -60  1.05  53   53
23    360  248.5  10.52  310  -125 1.05  31   31        ;
Table BD(i,*) Demands of each bus in MW
       Pd   Qd
*  removed for saving space;
table LN(i,j,*)    Network technical characteristics
                 r        x        b       limit
*  removed for saving space;
Table WD(t,*)
       w                    d
*  removed for saving space ;
parameter Wcap(i)
/8    200
19   150
21   100/;
LN(i,j,'x')$(LN(i,j,'x')=0)=LN(j,i,'x');LN(i,j,'r')$(LN(i,j,'r')
     =0)=LN(j,i,'r');
LN(i,j,'b')$(LN(i,j,'b')=0)=LN(j,i,'b');LN(i,j,'bij')$LN(i,j,'
     Limit')=1/LN(i,j,'x');
LN(i,j,'Limit')$(LN(i,j,'Limit')=0)=LN(j,i,'Limit');
LN(i,j,'z')$LN(i,j,'Limit')=sqrt(power(LN(i,j,'x'),2)+power(LN(i,
     j,'r'),2));
LN(j,i,'z')$(LN(i,j,'z')=0)=LN(i,j,'z');
LN(i,j,'th')$(LN(i,j,'Limit') and  LN(i,j,'x') and LN(i,j,'r'))
                                       =arctan(LN(i,j,'x')/(LN(i,j,'r')))
                                           ;
LN(i,j,'th')$(LN(i,j,'Limit') and  LN(i,j,'x') and LN(i,j,'r')=0)
      =pi/2;
LN(i,j,'th')$(LN(i,j,'Limit') and  LN(i,j,'r') and LN(i,j,'x')=0)
      =0;
LN(j,i,'th')$LN(i,j,'Limit')=LN(i,j,'th'); Parameter cx(i,j);
cx(i,j)$(LN(i,j,'limit')and LN(j,i,'limit'))=1; cx(i,j)$(cx(j,i))
     =1;
Variables OF,Pij(i,j,t),Qij(i,j,t),Pg(i,t),Qg(i,t),Va(i,t),V(i,t)
     ,Pw(i,t);
Equations eq1,eq2,eq3,eq4,eq5,eq6,eq7;
eq1(i,j,t)$cx(i,j).. Pij(i,j,t)=e=(V(i,t)*V(i,t)*cos(LN(j,i,'th'))
                    -V(i,t)*V(j,t)*cos(Va(i,t)-Va(j,t)+LN(j,i,'th'))
                        )/LN(j,i,'z');
```

```
eq2(i,j,t)$cx(i,j).. Qij(i,j,t)=e=(V(i,t)*V(i,t)*sin(LN(j,i,'th'))
                      -V(i,t)*V(j,t)*sin(Va(i,t)-Va(j,t)+LN(j,i,'th'))
                         )/LN(j,i,'z')
                                                    -LN(j,i,'b')*V(
                                                        i,t)*V(i,t)
                                                        /2;
eq3(i,t) .. Pw(i,t)$Wcap(i)+Pg(i,t)$GenD(i,'Pmax')-WD(t,'d')*BD(i
    ,'pd')/Sbase=e=
+sum(j$cx(j,i),Pij(i,j,t));
eq4(i,t) ..                Qg(i,t)$GenD(i,'Qmax')-WD(t,'d')*BD(i,'
    qd')/Sbase=e=
+sum(j$cx(j,i),Qij(i,j,t));
eq5          .. OF=g=sum((i,t),Pg(i,t)*GenD(i,'b')*Sbase$GenD(i,'
    Pmax'));
eq6(i,t)$(GenD(i,'Pmax') and ord(t)>1) ..Pg(i,t)-Pg(i,t-1)=l=GenD
    (i,'RU')/Sbase;
eq7(i,t)$(GenD(i,'Pmax') and ord(t)<card(t)) ..Pg(i,t)-Pg(i,t+1)=
    l=
                                               GenD(i,'RD')/Sbase;
Model loadflow     /eq1,eq2,eq3,eq4,eq5,eq6,eq7/;
Pg.lo(i,t)=GenD(i,'Pmin')/Sbase; Pg.up(i,t)=GenD(i,'Pmax')/Sbase;
Qg.lo(i,t)=GenD(i,'Qmin')/Sbase; Qg.up(i,t)=GenD(i,'Qmax')/Sbase;
Va.up(i,t)=pi/2; Va.lo(i,t)=-pi/2; Va.l(i,t)=0; Va.fx(slack,t)=0;
 Pij.up(i,j,t)$((cx(i,j)))=+1*LN(i,j,'Limit')/Sbase;
 Pij.lo(i,j,t)$((cx(i,j)))=-1*LN(i,j,'Limit')/Sbase;
 Qij.up(i,j,t)$((cx(i,j)))=+1*LN(i,j,'Limit')/Sbase;
 Qij.lo(i,j,t)$((cx(i,j)))=-1*LN(i,j,'Limit')/Sbase; V.lo(i,t)
    =0.9; V.up(i,t)=1.1;
Pw.up(i,t)=WD(t,'d')*Wcap(i)/sbase; Pw.lo(i,t)=0;
Solve loadflow minimizing OF using nlp;
```

Nomenclature

Parameters

$w_{i,t}$	Availability of wind turbine connected to bus *i* at time *t*
Λ_i^w	Capacity of wind turbine connected to bus *i*
$P_{i,t}^L$	Active power component of demand in bus *i* at time *t*
$P_{i,t}^w$	Active power generation by wind turbine connected to bus *i* at time *t*
$Q_{i,t}^w$	Reactive power generation by wind turbine connected to bus *i* at time *t*
a_g, b_g, c_g	Fuel cost coefficients of thermal unit *g*.
$P_i^{g,\max/\min}$	Maximum/minimum limits of power generation of thermal unit *g* connected to bus *i*.
$P_{ij}^{\max}$	Maximum power flow limits of branch connecting bus *i* to *j*.
x_{ij}	Reactance of branch connecting bus *i* to *j*.
r_{ij}	Resistance of branch connecting bus *i* to *j*.
VOLL	Value of loss of load ($/MW h)

VWC	Value of loss of wind ($/MW h)
b	Total line charging susceptance of branch connecting bus i to j (pu).

Variables

$S_{ij,t}$	Apparent power flow of branch connecting bus i to j at time t.
$P^g_{i,t}$	Active power generated by thermal unit g connected to bus i at time t (MW).
$P^w_{i,t}$	Active power generated by wind turbine connected to bus i at time t (MW).
$\lambda^p_{i,t}$	Active locational marginal price (LMP) in bus i at time t ($/MW h).
$P^{\text{LS}}_{i,t}$	Active Load shedding in bus i at time t (MW).
$P^{\text{wc}}_{i,t}$	Curtailed power of wind turbine connected to bus i at time t (MW).
$I_{ij,t}$	Current flow of branch connecting bus i to j at time t.
$Q^{\text{LS}}_{i,t}$	Reactive Load shedding in bus i at time t (MVAr).
$Q^g_{i,t}$	Reactive power generated by thermal unit g connected to bus i at time t (MW).
$\lambda^q_{i,t}$	Reactive locational marginal price (LMP) in bus i at time t ($/MVArh).
OF	Total operating costs ($).
$\delta_{i,t}$	Voltage angle in bus i at time t (rad).
$V_{i,t}$	Voltage magnitude in bus i at time t (pu).

References

1. A. Meeraus A. Brooke, D. Kendrick, R. Raman, *GAMS/Cplex 7.0 User Notes*. GAMS Development Corp. (2000)
2. A.J. Wood, B.F. Wollenberg, *Power Generation, Operation, and Control* (Wiley, Hoboken, 2012)
3. P. Maghouli, A. Soroudi, A. Keane, Robust computational framework for mid-term techno-economical assessment of energy storage. IET Gener. Transm. Distrib. **10**(3), 822–831 (2016)
4. F. Li, R. Bo Small test systems for power system economic studies, in: *IEEE PES General Meeting* (2010), pp. 1–4
5. R.D. Zimmerman, C.E. Murillo-Sanchez, R.J. Thomas, Matpower: steady-state operations, planning, and analysis tools for power systems research and education. IEEE Trans. Power Syst. **26**(1), 12–19 (2011)
6. A.J. Conejo, M. Carrión, J.M. Morales, *Decision Making Under Uncertainty in Electricity Markets*, vol. 1 (Springer, New York, 2010)
7. F. Bouffard, F.D. Galiana, A.J. Conejo, Market-clearing with stochastic security-part II: case studies. IEEE Trans. Power Syst. **20**(4), 1827–1835 (2005)